

# GRAPHIC DESIGN

In this new series, we look at some of the interesting routines that can be created from a few lines of BASIC code. These articles are intended to give you some ideas and start you off on your own programming projects. In this instalment, we see how mathematical graphs can be turned into intriguing designs with simple instructions.

Drawing a graph of a mathematical expression is easy enough. You simply take a range of values for one of the variables in the expression and work out the corresponding values for the other variable. With a simple graph such as  $Y = X^2$ , we could work out a table like this:

X	-5	-4	-3	-2	-1	0	1	2	3	4	5
Y	25	16	9	4	1	0	1	4	9	16	25

Drawing the points on a piece of squared paper and then joining them up with a smooth curve gives us the familiar necklace-shaped curve. The curve is the graph of  $Y=X^2$ . Another way of thinking of the curve is that it is the path traced out by a point that is moving along  $Y=X^2$ . The path is often called a *locus* and by manipulating and combining different loci, we can produce stunning patterns with the minimum of effort.

To do this, let's look at a well-known locus — the circle. The best way to describe a circle with a

formula is slightly more complicated than the method used above. Both the X and Y parts of the equation are defined in terms of a third variable or parameter. By varying this parameter through a given set of values, pairs of Xs and Ys can be worked out. The circle is given by the following equation:


$$X = R \sin (I)$$

$$Y = R \cos (I)$$

If we work out a set of values for X and Y as the angle I is turned a full circle (from 0 to 360°), we get the locus of a circle. The following short program for the Spectrum will create a circle. (See the 'Basic Flavours' box for other BASIC versions.)

```
10 REM Circle plot
20 LET xm=256: LET ym=176: LET xc=INT
(xm/2): LET yc=INT (ym/2)
30 LET r=50
40 LET s=PI/20
50 FOR i=0 TO 2*PI STEP s
60 INK 2: PLOT xc+r*SIN (i),yc+r*COS
(i)
70 NEXT i
```

Note that by changing the STEP size between the points plotted, you can vary the definition of the circle and alter the speed at which it is drawn. Most dialects of BASIC are not fast enough to draw smooth circles from many individual dots at an acceptable speed. To overcome this, it is often preferable to use a number of straight lines to connect the points of the circle. With a large number of straight lines, you can achieve a reasonable compromise between the speed and smoothness of the drawing. You can also use this formula to draw arcs and ellipses. For arcs, you select different values of I. For ellipses, you give R a different value in the X formula than in the Y formula. However, the circle is an adequate basis from which to create some interesting patterns.




Our first pattern is created by drawing a line from a fixed point to every point we plot on the locus. The following programs will position the point in the centre of the circle and to the circle's left respectively.



The next step is to use a moving point rather than a fixed one. We will plot two loci simultaneously and draw lines to connect corresponding points on the two paths. The simplest design this will create is two nested circles linked by numerous straight lines.

```
10 REM flower
20 LET xm=256: LET ym=176: LET xc=INT
(xm/2): LET yc=INT (ym/2)
30 LET r=50
40 LET s=PI/20
50 FOR i=0 TO 2*PI STEP s
60 INK 2: PLOT xc+r*SIN (i),yc+r*COS
(i): DRAW xc-(xc+r*SIN (i)),yc-(yc+r*COS
(i))
70 NEXT i
```

```
10 REM Nested circles
20 LET xm=256: LET ym=176: LET xc=INT
(xm/2): LET yc=INT (ym/2)
30 LET r=50
40 LET s=PI/20
50 FOR i=0 TO 2*PI STEP s
60 LET x=xc+(r+30)*SIN (i): LET y=yc+
(r+30)*COS (i)
70 LET p=xc+r*SIN (i): LET q=yc+r*COS
(i)
80 INK 2: PLOT x,y: DRAW p-x,q-y
90 NEXT i
```



```
10 REM Circle & Fixed point
20 LET xm=256: LET ym=176: LET xc=INT
(xm/2): LET yc=INT (ym/2)
30 LET r=50
40 LET s=PI/20
50 FOR i=0 TO 2*PI STEP s
60 INK 2: PLOT xc+r*SIN (i),yc+r*COS
(i): DRAW xc-100-(xc+r*SIN (i)),yc-(yc+r
*COS (i))
70 NEXT i
```

